
// MENULOG.WCM - Add and dynamically update 'Log / Resume' menu
//**********************************************************
// Author: Paul H. Caron
// Written for: WordPerfect 5.2 for OS/2 & Windows
// Version: 1.0
// Revision History: See README.WP

Application (WP;WPWP;Default)
ONCANCEL(Quit@)
MacroStatusPrompt(On!;"Updating WordPerfect's File Menu.  Please wait...")
LogOption:=""
ResumeOption:=""
MenuOption:=""
Call(ReadINI@)
If (ResumeOption="On")

Resume:="&Resume Profile..."
Else

Resume:="&Resume Profile... - Express"
EndIf
If (LogOption="On")

Log:="&Log Documents..."
Else

Log:="&Log Documents... - Express"
EndIf

If(MenuOption="No")
ONERROR(Quit@)
MenuDeleteItem("File Menu"; "Log / Resume")
Go(Quit@)

EndIf

//**************
Label(AddItems@)
ONERROR (DeleteItems@)
MenuAddItem("File Menu"; "Log / Resume";  7;  SubMenu!; "Log &/ Resume"; 

"Log-Resume documents")
MenuAddItem("Log / Resume";Log;21;Macro!;Log;"logfiles.wcm";

"Log selected documents in a user-defined Work Profile.")
MenuAddItem("Log / Resume";Resume;22;Macro!;Resume;"resume.wcm";

"Resume a previously logged Work Profile.")
MenuAddItem("Log / Resume";"LogState...";23;Macro!;"Log&State...";

"logstate.wcm";"Configure LogFiles and Resume.")
Go(Quit@)

//****************
Label(DeleteItems@)



ONERROR(AddItems@)
MenuDeleteItem("File Menu"; "Log / Resume")

Go(AddItems@)

//*************
Label(ReadINI@)

DLLLoad(MCRLink; "WPCMCR.DLL")
DLLCall(MCRLink;
"MCRGetIniString";

IniResult: WORD;{
"WPWP";
Ansistring ("Log-Resume Option");
Ansistring ("ResumePrompt");
Ansistring("On");
Address(ResumeOption)})

DLLCall(MCRLink;
"MCRGetIniString";

IniResult: WORD;{
"WPWP";
Ansistring ("Log-Resume Option");
Ansistring ("LogFilesPrompt");
Ansistring("On");
Address(LogOption)})

DLLCall(MCRLink;
"MCRGetIniString";

IniResult: WORD;{
"WPWP";
Ansistring ("Log-Resume Option");
Ansistring ("MenuOption");
Ansistring("No");
Address(MenuOption)})

DLLFREE(MCRLink)
Return

//**********
Label(Quit@)

Return


